

# Multivibrator/Timer CAD

National Semiconductor  
Application Brief 7  
March 1983




Circuit design making use of monolithic multivibrators and timers can be most easily and quickly done making use of a simple CAD (computer aided design) program. Fortunately, only 2 basic multivibrator types and 1 timer design type exist, reducing the program requirements to 3 sets of algorithms.

Figure 1 provides a view of the basic multivibrator and the  $t_{ON}$  (pulse width) determining resistor and capacitor (R1 and C), the Advanced Bipolar Logic Databook by National Semiconductor Corporation should be consulted for the specific device pinout and functions.

Figure 2 is a block diagram showing the basic timer and the  $t_{ON}$  (pulse width) determining resistor and capacitor (R1 and C) along with the  $t_{OFF}$  determining resistor R2 (required if

astable operation is required). The Linear Databook by National Semiconductor should be consulted for the specific device pinout and additional device functions.


This program is in "transportable" BASIC. No INPUT prompts for strings or ELSE statements are used. All IF statements, if true, result in a GOTO. Variable and array names are unique and are limited to 2 characters in length. This program consists of multiple statement lines, which must be dissected for some unexpanded microcomputers. Some microcomputers may require that the DATA statements on lines 850 and 860 be implemented as strings, and the READ statement on line 250 be replaced with the string handling routine.



TL/H/5248-1

Multivibrators accommodated by this program are shown in lines 130 through 200 of the listing.

FIGURE 1



TL/H/5248-2

The timers accommodated by this program are shown on lines 210 and 220 of the listing.

FIGURE 2

```

100 For I=0 to 10:PRINT:NEXT
110 PRINT"MULTIV by Bob Nelson — 2/1/83":PRINT:PRINT
120 PRINT"The following multivibrators are available for design:":Print
130 PRINT"DM54/74121      One Shots          0"
140 PRINT"DM54/74LS122   Retriggerable One Shots with Clear  1"
150 PRINT"DM54/74123     Dual Retriggerable One Shots with Clear  2"
160 PRINT"DM54/74L123   Dual Retriggerable One Shots with Clear  3"
170 PRINT"DM54/74LS123  Dual Retriggerable One Shots with Clear  4"
180 PRINT"DM54/74LS221  Dual One Shots with Schmitt-Trigger Inputs  5"
190 PRINT"DM86/9601     Retriggerable One Shots          6"
200 PRINT"DM86/9602     Dual Retriggerable, Resettable One Shots  7"
210 PRINT"LM555/555C    Timer                          8"
220 PRINT"LM556/556C    Dual Timer                        9"
230 PRINT:INPUT"Enter number representing choice . . . . .",N:PRINT
240 IF N>9 THEN 100
250 FOR I=0 TO N:READ K:READ RN:READ RX:NEXT:IF K<>.693 THEN 270
260 PRINT"Astable or Monostable (A/M)":INPUT M$
270 INPUT"Ton in nanoSeconds";T1:IF M$<>"A" THEN 300
280 INPUT"TOff in nanoSeconds";T2:IF 2*T2=<T1 THEN 300
290 PRINT"TOff cannot exceed 50% of Ton":PRINT:GOTO 270
300 INPUT"R1 in Kohms";R1:GOSUB 740:IF M$<>"A" THEN 320
310 INPUT"R2 in Kohms";R2
320 IF T1>0 AND R1>0 THEN 340
    
```

```

330 INPUT"C in picoFarads";C
340 IF K = .693 THEN 530
350 IF T1 > 0 AND R1 > 0 THEN 490
360 IF T1 > 0 AND C > 0 THEN 450
370 IF R1 > 0 AND C > 0 THEN 410
380 IF R1 > 0 THEN 400
390 R1 = RN:GOSUB 730:GOTO 350
400 GOSUB 710:GOTO 690
410 IF K = 0 THEN 430
420 T1 = K*C*(R1 + .7):GOTO 440
430 T1 = .7*C*R1
440 GOSUB 720:GOTO 690
450 IF K = 0 THEN 470
460 R1 = T1/(K*C -.7):GOTO 480
470 R1 = T1/(.7*C)
480 GOSUB 730:GOTO 690
490 IF K = 0 THEN 510
500 C = T1/(K*(R1 + .7)):GOTO 520
510 O = T1/(.7*R1)
520 GOSUB 790:GOTO 690
530 IF T1 = 0 THEN 570
540 IF T2 > 0 AND R1 > 0 AND R2 > 0 THEN 680
550 IF R1 > 0 AND T2 = 0 AND R2 = 0 THEN 680
560 IF C > 0 THEN 650
570 IF R1 = 0 THEN 600
580 IF C > 0 THEN 620
590 IF R1 > 0 THEN 610
600 R1 = RN:GOSUB 730:GOTO 530
610 GOSUB 710
620 T1 = K*(R1 + 2*R2)*C:T2 = K*R2*C:GOSUB 720:IF M$ <> "A" THEN 640
630 PRINT" Toff = ";T2;"nS"
640 GOTO 690
650 R2 = T2/(K*C):R1 = (T1/(K*C)) - 2*R2:GOSUB 730:IF M$ <> "A" THEN 670
660 PRINT"R2 = ";R2;"Kohms"
670 GOTO 690
680 C = (T1 + T2)/(K*(R1 + 2*R2)):GOSUB 790
690 PRINT:PRINT"Try same part again(Y/N)";:INPUT A$:IF A$ <> "N" THEN 270
700 END
710 PRINT"*****INSUFFICIENT DATA*****": RETURN
720 PRINT"Ton = ";T1;"nS":RETURN
730 PRINT"R1 = ";R1;"Kohms"
740 IF R1 >= RN THEN 760
750 PRINT"***** ??STABILITY - R1 <";RN;"Kohms *****"
760 IF R1 = <RX THEN 780
770 PRINT"***** ??ACCURACY - R1 >";RX;"Kohms *****"
780 RETURN
790 PRINT"C = ";C;"pF"
800 IF C > 0 THEN GOTO 820
810 GOSUB 710:GOTO 840
820 IF C > 1000 THEN 840
830 PRINT"***** ??ACCURACY - C < 1000 Pf *****"
840 RETURN
850 DATA 0, 1.4, 40, .45, 5, 260, .32, 5, 50, .33, 5, 400, .45, 5, 260
860 DATA .45, 1.4, 100, .31, 5, 50, .32, 5, 50, .693, .3, 10000, .693, .3, 10000

```



**LIFE SUPPORT POLICY**

NATIONAL'S PRODUCTS ARE NOT AUTHORIZED FOR USE AS CRITICAL COMPONENTS IN LIFE SUPPORT DEVICES OR SYSTEMS WITHOUT THE EXPRESS WRITTEN APPROVAL OF THE PRESIDENT OF NATIONAL SEMICONDUCTOR CORPORATION. As used herein:

1. Life support devices or systems are devices or systems which, (a) are intended for surgical implant into the body, or (b) support or sustain life, and whose failure to perform, when properly used in accordance with instructions for use provided in the labeling, can be reasonably expected to result in a significant injury to the user.
2. A critical component is any component of a life support device or system whose failure to perform can be reasonably expected to cause the failure of the life support device or system, or to affect its safety or effectiveness.



**National Semiconductor Corporation**  
1111 West Bardin Road  
Arlington, TX 76017  
Tel: 1(800) 272-9959  
Fax: 1(800) 737-7018

**National Semiconductor Europe**  
Fax: (+49) 0-180-530 85 86  
Email: onjwge@tevm2.nsc.com  
Deutsch Tel: (+49) 0-180-530 85 85  
English Tel: (+49) 0-180-532 78 32  
Français Tel: (+49) 0-180-532 93 58  
Italiano Tel: (+49) 0-180-534 16 80

**National Semiconductor Hong Kong Ltd.**  
19th Floor, Straight Block,  
Ocean Centre, 5 Canton Rd.  
Tsimshatsui, Kowloon  
Hong Kong  
Tel: (852) 2737-1600  
Fax: (852) 2736-9960

**National Semiconductor Japan Ltd.**  
Tel: 81-043-299-2309  
Fax: 81-043-299-2408